# Maintainer: Thomas Bächler <thomas@archlinux.org>

pkgname=iw
pkgver=5.19
pkgrel=2
pkgdesc='nl80211 based CLI configuration utility for wireless devices'
arch=('x86_64')
url='https://wireless.wiki.kernel.org/en/users/documentation/iw'
license=('GPL')
depends=('libnl')
makedepends=('linux-api-headers')
validpgpkeys=('C0EBC440F6DA091C884D8532E0F373F37BF9099A') # Johannes Berg
source=("https://www.kernel.org/pub/software/network/$pkgname/$pkgname-$pkgver.tar."{xz,sign})
sha256sums=('f167bbe947dd53bb9ebc0c1dcef5db6ad73ac1d6084f2c6f9376c5c360cc4d4e'
            'SKIP')

build() {
  cd "$srcdir"/$pkgname-$pkgver
  make
}

package() {
  cd "$srcdir"/$pkgname-$pkgver
  make DESTDIR="$pkgdir" SBINDIR="/usr/bin" install
}
